

XBasic Character Map Arrays

XBasic now supports a "character map array" for *GraphicsDesigner* and *GuiDesigner* grids. *GraphicsDesigner* functions that support this array are **XgrGetCharacterMapArray()** and **XgrSetCharacterMapArray()**. To install a character map array, call:

```
XgrSetCharacterMapArray (grid, @map[])
```

The **map[]** array must be an **XLONG** array with at least 256 elements, one for each 8-bit character. Someday, though not today, 65536 element character map arrays will be supported to deal with 16-bit character sets.

When a character map array is installed for a grid, the map array specifies the character *GraphicsDesigner* will draw for each character in the text string. The following code creates a character map array that maps every character into itself. Installing this character map array therefore does not change what is displayed.

```
DIM map[255]  
FOR i = 0 TO 255  
    map[i] = i  
NEXT i
```

To change what characters are displayed for certain characters, you simply change their entries in the character map array before you install it. For example:

```
map['a'] = 0x84      ' change 'a' to euro-a  
map['e'] = 0x89      ' change 'e' to euro-e  
map['i'] = 0x8C      ' change 'i' to euro-i  
map['o'] = 0x94      ' change 'o' to euro-o  
map['u'] = 0x81      ' change 'u' to euro-u  
map['y'] = 0x98      ' change 'y' to euro-y  
map['?'] = 0xA8      ' change '?' to inverted ?  
map['!'] = 0xAD      ' change '!' to inverted !  
XgrSetCharacterMapArray (grid, @map[])
```

You get the idea, right? After this code is executed, every 'a' in text drawn in the specified grid will be translated into what I stupidly call a "euro-a".

This method is fine if you want to change the character map for a few grids. To make EVERY grid in the environment and otherwise display the same mapped characters, simply install the character map array in grid # 0, as in:

```
XgrSetCharacterMapArray (0, @map[])
```

To make this happen automatically every time you start *XBasic*, simply put a "charmap.xxx" file in your working directory. If *XBasic* finds this file when it starts up, and the file contains at least 1024 bytes (the size of a 256 element **XLONG** array), the array is loaded and installed into grid # 0 before any grids are created. You are free to change the character map array of any grid at any time, including grid 0.

To create this file (after you've created the array as shown above), just put the following code after the code that defines the array.

```
ofile = OPEN ("charmap.xxx", $$WRNEW)
IF (ofile > 0) THEN
  WRITE [ofile], map[]
  CLOSE (ofile)
END IF
```

Nothing above mentions *GuiDesigner*. *GuiDesigner* programs send **#GetCharacterMapArray** and **#SetCharacterMapArray** messages to grids to accomplish the same thing, as in:

```
XuiSendMessage (grid, #GetCharacterMapArray, 0, 0, 0, 0, kid, @map[])
XuiSendMessage (grid, #SetCharacterMapArray, 0, 0, 0, 0, kid, @map[])
```

One thing though - **XuiSendMessage()** makes sure the specified grid argument is valid and the specified grid exists. Since grid # 0 is NEVER a valid grid number, you cannot set the global character map array with these messages. To do that you must call **XgrSetCharacterMapArray()**, which is written to understand this particular special meaning of grid # 0.